Step-by-step explanation of FEM with second order elements in *Mathematica*

José Luis Gómez Muñoz
http://homepage.cem.itesm.mx/jose.luis.gomez

Based on the work of John H. Mathews
http://math.fullerton.edu/mathews/n2003/GalerkinMod.html
and Joseph E. Flaherty
http://www.cs.rpi.edu/~flaherje/feaframe.html

A FEM approximation to the solution of an IVP, $-\frac{d}{dx}\left(p(x)\frac{d}{dx}y(x)\right)+q(x)\,y(x)=f(x)$, $a\le x\le b$, $y(a)=y_a$, $y(b)=y_b$, will be obtained. First the definition of the IVP for this example:

In[1]:=

```
Clear[x, y, p, q, f, a, b];

p[x_] := 1 + x;
q[x_] := 0;
f[x_] := x;
a = 0.0;
b = 5.0;
ya = -1.0;
yb = +1.0;

Print[TraditionalForm[-∂x (p[x] ∂x y[x]) + q[x] * y[x] == f[x]]];
Print[TraditionalForm[y[a] == ya]];
Print[TraditionalForm[y[b] == yb]];
```

$-(x+1)\,y''(x)-y'(x)=x$

$y(0.)=-1.$

$y(5.)=1.$

A sample domain partition for $a \le x \le b$:

In[12]:=

```
Clear[nele, j, coords];
nele = 4;
nver = nele + 1;
vertices = Table[a + (b - a) Log[j]/Log[nver], {j, 1, nver}]
```

Out[15]=

```
{0., 2.15338, 3.41303, 4.30677, 5.}
```

Each subdomain is an “element”, the endpoints of elements are “nodes”:

In[16]:=

```
distancia = (b - a)/10;
Graphics[{
  Thick,
  Table[{ColorData["NeonColors"][j / nele],

    Style[Text[j, {(vertices[[j]] + vertices[[j + 1]])/2, -distancia}, {0, -1}],
     Medium, Bold], Line[{{vertices[[j]], -distancia},
      {vertices[[j + 1]], -distancia}}]}, {j, 1, nele}],
  PointSize[Large], Table[Point[{vertices[[j]], -distancia}], {j, 1, nver}]},
 Frame → True, FrameTicks → {Automatic, None, Automatic, None}]
```

Out[17]=



More nodes must be inserted at each midpoint in order to do second order FEM:

In[18]:=

```
midpoints =
  Table[(vertices[[j]] + vertices[[j + 1]]) / 2, {j, 1, Length[vertices] - 1}];
coords = Riffle[vertices, midpoints]
```

Out[19]=

```
{0., 1.07669, 2.15338, 2.78321, 3.41303, 3.8599, 4.30677, 4.65338, 5.}
```

Elements and nodes together in the domain ("mesh"):

In[20]:=

```
nnod = Length[coords];
gnodos = {
   Thick,
   Table[{ColorData["NeonColors"][j / nele],
     Style[Text[j, {(vertices[[j]] + vertices[[j + 1]])/2, -distancia}, {0, -2}],
      Medium, Bold], Line[{{vertices[[j]], -distancia},
       {vertices[[j + 1]], -distancia}}]}, {j, 1, nele}],
   PointSize[Large],
   Table[{Style[Text[j, {coords[[j]], -distancia}, {0, 2}], Bold],
     Point[{coords[[j]], -distancia}]}, {j, 1, nnod}]};
Graphics[gnodos, Frame → True, FrameTicks → {Automatic, None, Automatic, None}]
```

Out[22]=



List of nodes that belong to each element:

In[23]:=

```
incidentes = Table[{j, j + 1, j + 2}, {j, 1, nnod - 2, 2}]
```

Out[23]=

```
{{1, 2, 3}, {3, 4, 5}, {5, 6, 7}, {7, 8, 9}}
```

For example, the nodes of the third element are:

In[24]:=

```
incidentes[[3]]
```

Out[24]=

```
{5, 6, 7}
```

The approximated solution $\Phi(x)$ will be a linear combination of some basis functions: $\Phi(x) = \sum_{j=1}^{n} c_j\,\phi_j(x)$. Here we have the usual definition of the basis functions for second order FEM in one dimention:

In[25]:=

```
Clear[ϕ, x];
Subscript[ϕ, 1][x_] :=
  ⎧ 1 - 3 ((x-coords[[1]])/(coords[[3]]-coords[[1]])) + 2 ((x-coords[[1]])/(coords[[3]]-coords[[1]]))^2   coords[[1]] < x ≤ coords[[3]]
  ⎩ 0                                                                                                       True ;
Subscript[ϕ, nnod][x_] :=
  ⎧ 1 + 3 ((x-coords[[nnod]])/(coords[[nnod]]-coords[[nnod-2]])) +       coords[[nnod - 2]] < x ≤ coords[[nno
  ⎨   2 ((x-coords[[nnod]])/(coords[[nnod]]-coords[[nnod-2]]))^2
  ⎩ 0                                                                    True
  ;
Subscript[ϕ, j_?EvenQ][x_] :=
  ⎧ 1 - 4 ((x-coords[[j]])/(coords[[j+1]]-coords[[j-1]]))^2   coords[[j - 1]] < x ≤ coords[[j + 1]]
  ⎩ 0                                                          True ;
Subscript[ϕ, j_?OddQ][x_] :=
  ⎧ 1 + 3 ((x-coords[[j]])/(coords[[j]]-coords[[j-2]])) +        coords[[j - 2]] < x ≤ coords[[
  ⎪   2 ((x-coords[[j]])/(coords[[j]]-coords[[j-2]]))^2
  ⎨ 1 - 3 ((x-coords[[j]])/(coords[[j+2]]-coords[[j]])) +        coords[[j]] < x ≤ coords[[j + :
  ⎪   2 ((x-coords[[j]])/(coords[[j+2]]-coords[[j]]))^2
  ⎩ 0                                                            True
  ;
funciones = Table[ϕ_j[x], {j, 1, nnod}];
Table[Plot[funciones[[j]], {x, a, b},
  PlotRange → {-2 distancia, 1.2}, PlotStyle → {Red, Thick}, Epilog → gnodos,
  Axes → False, Frame → True, PlotLabel → φ_j[x]], {j, 1, nnod}]
```

Out[31]=



Basis functions and domain meshing:

In[32]:=

```
Plot[funciones, {x, a, b}, PlotRange → {-2 distancia, 1 + distancia},
 Frame → True, Axes → False, Epilog → gnodos]
```

Out[32]=



Approximated solution $\Phi(x) = \sum_{j=1}^{n} c_j\, \phi_j(x)$. The goal is to find c_j such that $\Phi(x)$ is (in some sense) the best approximation:

In[33]:=

```
Clear[Φ, c];

Φ[x_] := Sum[c_j * ϕ_j[x], {j, 1, nnod}];

Print["\nApproximated solution with parameters c_j:\n"];

Print["Φ[x]=", TraditionalForm[Φ[x]]];

Print["\nWe will find c_j such that
    Φ(x) is (in some sense) the best approximation:\n"];
```

Approximated solution with parameters c_j:

$$\Phi[x]=c_1\left(\begin{cases}0.431308\,(x+0.)^2-1.39316\,(x+0.)+1 & 0.<x\le 2.15338\\ 0 & \text{True}\end{cases}\right)+$$
$$c_2\left(\begin{cases}1-0.862616\,(x-1.07669)^2 & 0.<x\le 2.15338\\ 0 & \text{True}\end{cases}\right)+$$
$$c_4\left(\begin{cases}1-2.52093\,(x-2.78321)^2 & 2.15338<x\le 3.41303\\ 0 & \text{True}\end{cases}\right)+$$
$$c_6\left(\begin{cases}1-5.00775\,(x-3.8599)^2 & 3.41303<x\le 4.30677\\ 0 & \text{True}\end{cases}\right)+$$
$$c_8\left(\begin{cases}1-8.32338\,(x-4.65338)^2 & 4.30677<x\le 5.\\ 0 & \text{True}\end{cases}\right)+$$
$$c_9\left(\begin{cases}4.16169\,(x-5.)^2+4.32754\,(x-5.)+1 & 4.30677<x\le 5.\\ 0 & \text{True}\end{cases}\right)+$$
$$c_3\left(\begin{cases}0.431308\,(x-2.15338)^2+1.39316\,(x-2.15338)+1 & 0.<x\le 2.15338\\ 1.26047\,(x-2.15338)^2-2.38162\,(x-2.15338)+1 & 2.15338<x\le 3.41303\end{cases}\right)+$$
$$c_5\left(\begin{cases}1.26047\,(x-3.41303)^2+2.38162\,(x-3.41303)+1 & 2.15338<x\le 3.41303\\ 2.50388\,(x-3.41303)^2-3.3567\,(x-3.41303)+1 & 3.41303<x\le 4.30677\end{cases}\right)+$$
$$c_7\left(\begin{cases}2.50388\,(x-4.30677)^2+3.3567\,(x-4.30677)+1 & 3.41303<x\le 4.30677\\ 4.16169\,(x-4.30677)^2-4.32754\,(x-4.30677)+1 & 4.30677<x\le 5.\end{cases}\right)$$

We will find c_j such that Φ(x) is (in some sense) the best approximation:

Differential equation:

$$-\frac{d}{dx}\left(p(x)\,\frac{d}{dx}\,y(x)\right)+q(x)\,y(x)-f(x)=0$$

If we replace $\Phi(x)$ instead of $y(x)$ in the left-hand side, we obtain a residual, $r(x)$. instead of zero:

$$-\frac{d}{dx}\left(p(x)\,\frac{d}{dx}\,\Phi(x)\right)+q(x)\,\Phi(x)-f(x)=r(x)$$

Galerkin methods consist on findind $\Phi(x)=\sum_{j=1}^{n} c_j\,\phi_j(x)$ such that its residual $r(x)$ has a zero component in the space of functions $\phi_j(x)$. That means that the internal product (integral) of $r(x)$ with each $\phi_j(x)$ must be zero. In FEM, this calculation is performed element by element:

$$\int_{element}\phi_j(x)\,r(x)\,dx=0$$

that means:

$$-\int_{element}\phi_j(x)\,\frac{d}{dx}\left(p(x)\,\frac{d}{dx}\,\Phi(x)\right)dx+\int_{element}\phi_j(x)\,(q(x)\,\Phi(x)-f(x))\,dx=0$$

However our $\Phi(x)$ does not have second derivatives, therefore integration by parts is used and we obtain:

$$\int_{element}p(x)\,\frac{d\Phi(x)}{dx}\,\frac{d\phi_j(x)}{dx}\,dx+\int_{element}\phi_j(x)\,(q(x)\,\Phi(x)-f(x))\,dx=0$$

The evaluation of the integrals gives equations for the parameters c_j. Those integrals are evaluated using a “Gauss quadrature” in FEM:

In[38]:=

```
parametros = Table[c_j, {j, 1, nnod}];
matrices = {};
cargas = {};
integrationweights = N[{5/9, 8/9, 5/9}];

integrationpoints = N[{-Sqrt[3/5], 0, Sqrt[3/5]}];


Do[
 nodoini = First[incidentes[[elemento]] ];
 nodofin = Last[incidentes[[elemento]] ];

 xini = coords[[nodoini]];
 xfin = coords[[nodofin]];

 (* Below, ξ is a local variable inside the element, -1≤ξ≤1 *)
 integrationvar[ξ_] := (1 - ξ)/2 xini + (1 + ξ)/2 xfin ;

 integrationfunction[u_, k_] := p[u] * Φ'[u] * ϕ_k'[u] + ϕ_k[u] * (q[u] * Φ[u] - f[u]);

 ecuaciones =
  Table[
   0.5 * (xfin - xini) * Sum[integrationweights[[j]] *
       integrationfunction[integrationvar[integrationpoints[[j]] ], k],
      {j, 1, Length[integrationpoints]}] == 0,
   {k, 1, nnod}];

 ecuaciones = ReplaceAll[ecuaciones, True → 0];

 {v, m} = N[ CoefficientArrays[ecuaciones, parametros] ];

 AppendTo[matrices, m];
 AppendTo[cargas, -v];

 Print["Equations without boundary conditions, element:", elemento];
 Print[MatrixForm[m], ".",
  MatrixForm[parametros], "==",
  MatrixForm[-v]
 ],
 {elemento, 1, nele}]
```

Equations without boundary conditions, element:1

$$\begin{pmatrix} 1.58357 & -1.90503 & 0.321462 & 0. & 0. & 0. & 0. & 0. & 0. \\ -1.90503 & 5.14339 & -3.23836 & 0. & 0. & 0. & 0. & 0. & 0. \\ 0.321462 & -3.23836 & 2.9169 & 0. & 0. & 0. & 0. & 0. & 0. \\ 0. & 0. & 0. & 0. & 0. & 0. & 0. & 0. & 0. \\ 0. & 0. & 0. & 0. & 0. & 0. & 0. & 0. & 0. \\ 0. & 0. & 0. & 0. & 0. & 0. & 0. & 0. & 0. \\ 0. & 0. & 0. & 0. & 0. & 0. & 0. & 0. & 0. \\ 0. & 0. & 0. & 0. & 0. & 0. & 0. & 0. & 0. \\ 0. & 0. & 0. & 0. & 0. & 0. & 0. & 0. & 0. \end{pmatrix} . \begin{pmatrix} c_1 \\ c_2 \\ c_3 \\ c_4 \\ c_5 \\ c_6 \\ c_7 \\ c_8 \\ c_9 \end{pmatrix} == \begin{pmatrix} 4.16334\times 10^{-17} \\ 1.54569 \\ 0.772843 \\ 0. \\ 0. \\ 0. \\ 0. \\ 0. \\ 0. \end{pmatrix}$$

Equations without boundary conditions, element:2

$$\begin{pmatrix} 0. & 0. & 0. & 0. & 0. & 0. & 0. & 0. & 0. \\ 0. & 0. & 0. & 0. & 0. & 0. & 0. & 0. & 0. \\ 0. & 0. & 6.34123 & -7.34236 & 1.00113 & 0. & 0. & 0. & 0. \\ 0. & 0. & -7.34236 & 16.018 & -8.67569 & 0. & 0. & 0. & 0. \\ 0. & 0. & 1.00113 & -8.67569 & 7.67456 & 0. & 0. & 0. & 0. \\ 0. & 0. & 0. & 0. & 0. & 0. & 0. & 0. & 0. \\ 0. & 0. & 0. & 0. & 0. & 0. & 0. & 0. & 0. \\ 0. & 0. & 0. & 0. & 0. & 0. & 0. & 0. & 0. \\ 0. & 0. & 0. & 0. & 0. & 0. & 0. & 0. & 0. \end{pmatrix} . \begin{pmatrix} c_1 \\ c_2 \\ c_3 \\ c_4 \\ c_5 \\ c_6 \\ c_7 \\ c_8 \\ c_9 \end{pmatrix} == \begin{pmatrix} 0. \\ 0. \\ 0.452084 \\ 2.33724 \\ 0.716536 \\ 0. \\ 0. \\ 0. \\ 0. \end{pmatrix}$$

Equations without boundary conditions, element:3

$$\begin{pmatrix} 0. & 0. & 0. & 0. & 0. & 0. & 0. & 0. & 0. \\ 0. & 0. & 0. & 0. & 0. & 0. & 0. & 0. & 0. \\ 0. & 0. & 0. & 0. & 0. & 0. & 0. & 0. & 0. \\ 0. & 0. & 0. & 0. & 0. & 0. & 0. & 0. & 0. \\ 0. & 0. & 0. & 0. & 12.0214 & -13.834 & 1.81258 & 0. & 0. \\ 0. & 0. & 0. & 0. & -13.834 & 29.0013 & -15.1673 & 0. & 0. \\ 0. & 0. & 0. & 0. & 1.81258 & -15.1673 & 13.3547 & 0. & 0. \\ 0. & 0. & 0. & 0. & 0. & 0. & 0. & 0. & 0. \\ 0. & 0. & 0. & 0. & 0. & 0. & 0. & 0. & 0. \end{pmatrix} . \begin{pmatrix} c_1 \\ c_2 \\ c_3 \\ c_4 \\ c_5 \\ c_6 \\ c_7 \\ c_8 \\ c_9 \end{pmatrix} == \begin{pmatrix} 0. \\ 0. \\ 0. \\ 0. \\ 0.508391 \\ 2.29982 \\ 0.641518 \\ 0. \\ 0. \end{pmatrix}$$

Equations without boundary conditions, element:4

$$\begin{pmatrix} 0. & 0. & 0. & 0. & 0. & 0. & 0. & 0. & 0. \\ 0. & 0. & 0. & 0. & 0. & 0. & 0. & 0. & 0. \\ 0. & 0. & 0. & 0. & 0. & 0. & 0. & 0. & 0. \\ 0. & 0. & 0. & 0. & 0. & 0. & 0. & 0. & 0. \\ 0. & 0. & 0. & 0. & 0. & 0. & 0. & 0. & 0. \\ 0. & 0. & 0. & 0. & 0. & 0. & 0. & 0. & 0. \\ 0. & 0. & 0. & 0. & 0. & 0. & 18.3619 & -21.0802 & 2.71836 \\ 0. & 0. & 0. & 0. & 0. & 0. & -21.0802 & 43.4938 & -22.4135 \\ 0. & 0. & 0. & 0. & 0. & 0. & 2.71836 & -22.4135 & 19.6952 \end{pmatrix} . \begin{pmatrix} c_1 \\ c_2 \\ c_3 \\ c_4 \\ c_5 \\ c_6 \\ c_7 \\ c_8 \\ c_9 \end{pmatrix} == \begin{pmatrix} 0. \\ 0. \\ 0. \\ 0. \\ 0. \\ 0. \\ 0.4976 \\ 2.15059 \\ 0.577695 \end{pmatrix}$$

All the equations are added ("Assembly of the global matrix") in order to obtain a unique system of equations for the parameters c_j. This system is singular (it does not have a unique solution) because boundary conditions have not been inserted yet:

In[44]:=
```
matrizGlobal = Total[matrices];

cargaGlobal = Total[cargas];

Print["Equations without boundary conditions"];
Print[NumberForm[MatrixForm[matrizGlobal], 3], ".",
 MatrixForm[parametros], "==",
 NumberForm[MatrixForm[cargaGlobal], 3]
]
```

Equations without boundary conditions

$$\begin{pmatrix} 1.58 & -1.91 & 0.321 & 0. & 0. & 0. & 0. & 0. & 0. \\ -1.91 & 5.14 & -3.24 & 0. & 0. & 0. & 0. & 0. & 0. \\ 0.321 & -3.24 & 9.26 & -7.34 & 1. & 0. & 0. & 0. & 0. \\ 0. & 0. & -7.34 & 16. & -8.68 & 0. & 0. & 0. & 0. \\ 0. & 0. & 1. & -8.68 & 19.7 & -13.8 & 1.81 & 0. & 0. \\ 0. & 0. & 0. & 0. & -13.8 & 29. & -15.2 & 0. & 0. \\ 0. & 0. & 0. & 0. & 1.81 & -15.2 & 31.7 & -21.1 & 2.72 \\ 0. & 0. & 0. & 0. & 0. & 0. & -21.1 & 43.5 & -22.4 \\ 0. & 0. & 0. & 0. & 0. & 0. & 2.72 & -22.4 & 19.7 \end{pmatrix} . \begin{pmatrix} c_1 \\ c_2 \\ c_3 \\ c_4 \\ c_5 \\ c_6 \\ c_7 \\ c_8 \\ c_9 \end{pmatrix} == \begin{pmatrix} 4.16 \times 10^{-17} \\ 1.55 \\ 1.22 \\ 2.34 \\ 1.22 \\ 2.3 \\ 1.14 \\ 2.15 \\ 0.578 \end{pmatrix}$$

Boundary conditions are inserted; now the system has a unique solution:

In[48]:=
```
matrizGlobal[[1]] = Prepend[Table[0, {j, 2, nnod}], 1];
cargaGlobal[[1]] = ya;

matrizGlobal[[nnod]] = Append[Table[0, {j, 2, nnod}], 1];
cargaGlobal[[nnod]] = yb;

Print["Equations WITH boundary conditions"];
Print[NumberForm[MatrixForm[matrizGlobal], 3], ".",
 MatrixForm[parametros], "==",
 NumberForm[MatrixForm[cargaGlobal], 3]
]
```

Equations WITH boundary conditions

$$\begin{pmatrix} 1 & 0. & 0. & 0. & 0. & 0. & 0. & 0. & 0. \\ -1.91 & 5.14 & -3.24 & 0. & 0. & 0. & 0. & 0. & 0. \\ 0.321 & -3.24 & 9.26 & -7.34 & 1. & 0. & 0. & 0. & 0. \\ 0. & 0. & -7.34 & 16. & -8.68 & 0. & 0. & 0. & 0. \\ 0. & 0. & 1. & -8.68 & 19.7 & -13.8 & 1.81 & 0. & 0. \\ 0. & 0. & 0. & 0. & -13.8 & 29. & -15.2 & 0. & 0. \\ 0. & 0. & 0. & 0. & 1.81 & -15.2 & 31.7 & -21.1 & 2.72 \\ 0. & 0. & 0. & 0. & 0. & 0. & -21.1 & 43.5 & -22.4 \\ 0. & 0. & 0. & 0. & 0. & 0. & 0. & 0. & 1 \end{pmatrix} . \begin{pmatrix} c_1 \\ c_2 \\ c_3 \\ c_4 \\ c_5 \\ c_6 \\ c_7 \\ c_8 \\ c_9 \end{pmatrix} == \begin{pmatrix} -1. \\ 1.55 \\ 1.22 \\ 2.34 \\ 1.22 \\ 2.3 \\ 1.14 \\ 2.15 \\ 1. \end{pmatrix}$$

Solution of the system of equations:

In[54]:=

```
solvector = LinearSolve[matrizGlobal, cargaGlobal];
solu = Table[parametros[[j]] → solvector[[j]], {j, 1, nnod}]
```

Out[55]=

$\{c_1 \to -1., c_2 \to 1.56216, c_3 \to 2.5921, c_4 \to 2.71696, c_5 \to 2.55322, c_6 \to 2.27532, c_7 \to 1.87021, c_8 \to 1.47121, c_9 \to 1.\}$

The solution values are replaced in $\Phi(x) = \sum_{j=1}^{n} c_j \phi_j(x)$. The result is called g(x):

In[56]:=

```
g[x_] := ReplaceAll[Φ[x], solu ];

Print["FEM approximated solution"];
Print["g[x]=", TraditionalForm[g[x]]]
```

FEM approximated solution

$$\text{g[x]=}1. \left(\begin{cases} 4.16169\,(x-5.)^2 + 4.32754\,(x-5.) + 1 & 4.30677 < x \le 5. \\ 0 & \text{True} \end{cases} \right) +$$
$$1.47121 \left(\begin{cases} 1 - 8.32338\,(x-4.65338)^2 & 4.30677 < x \le 5. \\ 0 & \text{True} \end{cases} \right) +$$
$$2.27532 \left(\begin{cases} 1 - 5.00775\,(x-3.8599)^2 & 3.41303 < x \le 4.30677 \\ 0 & \text{True} \end{cases} \right) +$$
$$2.71696 \left(\begin{cases} 1 - 2.52093\,(x-2.78321)^2 & 2.15338 < x \le 3.41303 \\ 0 & \text{True} \end{cases} \right) +$$
$$1.56216 \left(\begin{cases} 1 - 0.862616\,(x-1.07669)^2 & 0. < x \le 2.15338 \\ 0 & \text{True} \end{cases} \right) -$$
$$1. \left(\begin{cases} 0.431308\,(x+0.)^2 - 1.39316\,(x+0.) + 1 & 0. < x \le 2.15338 \\ 0 & \text{True} \end{cases} \right) +$$
$$1.87021 \left(\begin{cases} 2.50388\,(x-4.30677)^2 + 3.3567\,(x-4.30677) + 1 & 3.41303 < x \le 4.30677 \\ 4.16169\,(x-4.30677)^2 - 4.32754\,(x-4.30677) + 1 & 4.30677 < x \le 5. \end{cases} \right) +$$
$$2.55322 \left(\begin{cases} 1.26047\,(x-3.41303)^2 + 2.38162\,(x-3.41303) + 1 & 2.15338 < x \le 3.41303 \\ 2.50388\,(x-3.41303)^2 - 3.3567\,(x-3.41303) + 1 & 3.41303 < x \le 4.30677 \end{cases} \right) +$$
$$2.5921 \left(\begin{cases} 0.431308\,(x-2.15338)^2 + 1.39316\,(x-2.15338) + 1 & 0. < x \le 2.15338 \\ 1.26047\,(x-2.15338)^2 - 2.38162\,(x-2.15338) + 1 & 2.15338 < x \le 3.41303 \end{cases} \right)$$

Graph of the approximated FEM solution:

In[59]:=

```
Plot[g[x], {x, a, b}, PlotLabel → "FEM solution"]
```

Out[59]=



Graph of the exact (analytical) solution:

In[60]:=

```
solexacta =
  DSolve[{-∂x (p[x] ∂x y[x]) + q[x] * y[x] == f[x], y[a] == ya, y[b] == yb}, y[x], x];

exacta[x_] := Evaluate[Expand[ReplaceAll[y[x], solexacta[[1]] ]] ];

Plot[exacta[x], {x, a, b}, PlotLabel → exacta[x]]
```

Out[62]=



Compare the FEM solution and the exact solution:

In[63]:=
```
Plot[{exacta[x], g[x]}, {x, a, b}, PlotLabel → "FEM and exact solutions"]
```

Out[63]=



Error:

In[64]:=
```
Plot[exacta[x] - g[x], {x, a, b}, PlotLabel → "Error = Exact-FEM"]
```

Out[64]=



Exercises

1. Modify this document to obtain a FEM solution to same problem, but using 6 elements of the same size; the nodes are equally spaced:


1
2
3
4
5
6
1 2 3 4 5 6 7 8 9 10 11 12 13
0 1 2 3 4 5




Solución aproximada


Error = Analitica − Galerkin

References

Based on the work of John H. Mathews
http://math.fullerton.edu/mathews/n2003/GalerkinMod.html
and Joseph E. Flaherty
http://www.cs.rpi.edu/~flaherje/feaframe.html

José Luis Gómez Muñoz

http://homepage.cem.itesm.mx/jose.luis.gomez

In[65]:=
```
$Version
```
Out[65]=
```
9.0 for Microsoft Windows (64-bit) (January 25, 2013)
```

In[66]:=
```
DateString[]
```
Out[66]=
```
Wed 2 Apr 2014 15:36:33
```